

COAL WASHERY SIMULATION  
A BASIC MICROCOMPUTER PROGRAM

Prof. D.D.Misra

Abstract

A number of elaborate and sophisticated programs have been developed for the simulation of coal preparation plants. However most of them are suitable for the main frame computers. This paper describes and gives the listing of a basic program suitable for personal computers and other microcomputer based systems. This highly simplified program is suitable for flow sheet calculations for the design of coal preparation plants as well as for teaching computer aided design (CAD) of such plants. Though the models used in this program are very much simplified and empirical in nature the results obtained are accurate enough within the limits of sampling errors normally encountered in the coal preparation plants. There is provision for addition of new subroutines for unit operations not already included and for elaboration of the models used if the memory size of the computer permits.

Introduction

Computer assistance for coal washery design has been in use for about two decades [1]. Gradually the programs have become progressively more elaborate as well as comprehensive. Such programs now include most of the density separation processes as well as the size reduction and size separation processes [2,3,4]. However because of their elaborate nature and the sophisticated and rigorous models used, most of the programs are suitable for the main frame computers. Lately, with the development and popularisation of a large variety of small personal computers, attempts are being made to modify the programs for use in these small computers. This paper describes a program written in BASIC for such use. It is important to note that because of the simplification desired and the limitation on the speed, as well as the time of execution, the approximate and empirical relationships have been preferred to the more sophisticated and rigorous models used in the programs for the main frame computers. The listing of the program is given in appendix I.

Data on feed coal

The data for the program on the feed coal consists of the following.

1. Size analysis of the feed coal: Upto 8 sieve sizes can be used for the program. The sieve sizes are to be given in millimeters.
2. Density analysis of each size fraction: Upto 10 densities can be used. The density values are to be given in grams per cubic centimeters.

The size-density analyses mentioned above generates an array which may be termed the "Size density distribution matrix" of the feed coal. As an example such a matrix for a particular feed coal is given in table 1. The computer, executing the program, computes, and holds in the memory, this matrix for each flow stream of the plant. The sum of all the elements of the size density distribution matrix equals the flow rate or yield of the flow stream.

Table 1: Size density distribution matrix of a coal sample

(Differential percentages retained and floating at :)								
(Sizes in millimeters and densities in grams/cubic cm.)								
Density	200.0	100.0	50.0	25.0	12.5	6.0	1.0	0.5
1.3	2.3	1.6	0.9	0.5	1.2	1.3	0.7	1.6
1.4	5.0	3.4	2.1	1.2	0.8	1.6	0.9	1.7
1.45	2.8	2.1	1.4	0.6	0.9	1.1	1.4	0.6
1.5	2.2	1.7	1.6	0.9	1.1	1.4	0.6	0.6
1.55	2.7	1.3	1.7	0.7	0.7	0.6	0.4	1.3
1.6	1.2	1.5	1.5	0.7	0.3	0.4	0.5	0.8
1.65	2.1	2.1	1.3	0.3	0.6	0.5	0.3	1.2
1.7	1.9	2.5	0.5	0.6	0.4	1.1	0.6	1.0
1.8	3.2	0.9	1.2	0.3	0.8	0.9	0.8	0.8
2.2	1.8	1.9	1.8	1.2	1.2	2.5	0.5	0.0
Total	25.2	19.0	14.0	7.0	8.0	11.0	6.0	9.8

3. Ash content of each size density fraction: If so desired sulphur content or any other assay value, in stead of the ash content, can be given. The computer calculates the ash factors (or assay factors) and holds them in the memory in the form of an array corresponding to the size density distribution matrix.

The nature of the flow rate, i.e. as mass per unit of time (such as tonnes per hour) or percentage, and the chemical analysis of interest, such as ash %, sulphur content etc. are to be specified in the data statements. For example in the version of the program given in appendix I the flow rate is given as "FLOW RATE AS % OF FEED" and the chemical analysis is given as "WITH ASH %" (See line number 11000). The total data on feed coal has to be given in the following manner.

1. Number of sieve sizes used for size analysis (maximum 8), number of density cuts made (maximum 10), the description of the flow and the description of the chemical analysis. The descriptions are reproduced by the program when the results are printed.
2. The sieve sizes in millimeters and the density values in gm/cc (See line numbers 11010 and 11020).
3. For each size the differential weight fractions (or the percentages or the tonnes per hour etc.) for each density fraction and then the ash content (or assay value) for the density fractions (See line numbers 11030 to 11180. In this particular example there are 8 sets of data each set having 10 weight percentages and 10 ash values).

#### Flow sheet data

Different forms of matrix representation of the plant flow sheet have been suggested to facilitate computer simulation of the plants [5,6]. A type of process matrix convention has been adopted for this program. The flow sheet data for the program is prepared as follows.

1. The flow sheet for the plant is drawn in the conventional manner. Each flow stream is given a separate identification number.

The maximum number of flow streams permissible with this program is 20. This number can be increased by suitably changing the dimensions of the arrays COAL and ASHF (See line number 30). Similarly the maximum number of nodes (unit operations) permissible with this program is limited by the dimensions of the arrays MFS and EQUIP. Here it is important to note that if there is a recycle stream an extra node usually referred to as the convergence block, has to be introduced and separate numbers have to be given to the input and output streams of the node.

2. The flow sheet matrix consists of twelve columns and number of rows equal to the number of nodes plus the rows required for printing the results. Correspondingly each element of the row has to be assigned the following values.

Element 1: Code number identifying the unit operation. The code numbers now used are given in appendix II. More code numbers can be added if subroutines for new unit operations are written.

Elements 2 to 4: Input stream numbers (except in case of the nodes for mixing or blending when elements 2 to 6 denote the input stream number and recycle node where element 4 denotes the serial number of the row to which recycle takes place).

Elements 5 to 7: Output stream numbers (except in case of the nodes for mixing or blending where the output stream number is denoted by element 7 and recycle node where element 7 gives the maximum number of iterations to be tried for convergence. If the number of iterations is given as zero it is taken as 50, the default value.

Elements 8 to 12: The equipment dimensions and the other parameters for the unit operations. Actually in the simplified version much less than the five parameters are sufficient. The data required for each type of equipment/unit operation are given in appendix II. If changes are made in the models included in the program these parameters may have to be changed.

3. For activating the subroutine that prints the result of the simulation a row with the first element zero has to be given in the flow sheet matrix. In that case the elements 2 to 7 of the row are taken as the numbers of the flow streams the flow rates (yields) of which are to be printed. Thus the yields (or flow rates) of six flow streams can be printed by one such row. When the number of flow streams, for which the results are to be printed, exceeds six, more such rows have to be given. For example in appendix I there are 20 flow streams and to print the 20 flow rates four rows have been included (See lines 11250, 11280, 11320 and 11350).
4. When the first element of a row is 10, the program prints the size density distribution of the flow streams, the numbers of which are given as elements 2 to 7. The print out is in the form of the array shown in table 1.
5. The total number of rows in the flow sheet matrix, including the recycle nodes, rows for printing the yields and the rows for printing the size density distribution matrices, if any, has to be given before the data on the flow sheet matrix.

#### Program monitoring

When the command for execution (RUN) of the program is given the video screen is cleared (See line 25) and the message "SIMULATION IN PROGRESS" is displayed. The computer then reads and echo prints the data given in the program. The echo prints help in identifying any error in the data. As the simulation proceeds the computer displays the position in the flow sheet by suitable messages such as "AT CRUSHER", "AT MIXING" etc. These displays help in locating errors that may stop execution. In addition there are some other error messages such as the Eckart probable in a density separation process becoming zero or negative. When the execution of the program is over the message "SIMULATION COMPLETED" is displayed (See line 305).

The data for rows to activate the printing of results can be given at the end of the flow sheet matrix. However it is better to give these rows immediately after the relevant unit operation rows so that at least a part of the simulation result may be available if the execution is interrupted.

## scope and limitations

The program has been written using a building block approach with one subroutine representing one unit operation in a coal preparation plant. This approach helps elaboration or modification of the program, if desired, by the adding new subroutines or improving the subroutines provided when the memory size of the computer permits. For example in the listing of appendix I line 230 provides for a subroutine starting at line 6000 and ending before line 7000. This can be used for adding a unit operation, such as froth flotation, which is now not provided for.

In case of the crushers it is assumed that a roll crusher would be used if the set is 37.5 mm or larger and a hammer mill or a similar crusher would be used if the set is smaller. The subroutine for the crusher does not incorporate any model for the liberation of the coal from the associated mineral matter. However it includes a simplified empirical model for differential breakage.

Another limitation is that no provision has been made for effect of the design and operating variables such as the number of rolls (single or double), speed of revolution etc. and the size distributions for certain standard conditions have been taken.

In case of the density separation processes the partition value curve, more commonly known as the Tromp curve, is the most useful tool for process simulation. This curve varies in nature from process to process [7]. The best way to simulate the density separation process is to take into account the characteristic curve of the process. Better accuracy is obtained by representing the curve by a set of discrete points and obtaining the intermediate points by a suitable procedure for interpolation, such as the Lagrangian interpolation [8]. However such discrete point representation requires more memory and the interpolation procedures are generally time taking. Therefore in this program a highly simplified empirical equation is used for the Tromp curves of all the processes (See lines 7075 and 7080). The differences between the processes are taken into account by the imperfection value and a parameter SK (See line 7040).

It must be noted here that though the models included in the program are highly simplified and empirical the results are reproducible within the limits of the sampling error that are usually encountered in coal washeries.

Some results of simulation

Appendix III gives the results of simulation of the flow sheet shown in figure 1 and the corresponding flow sheet matrix has been included in the listing of appendix I.

This flow sheet is one of the alternatives that were taken into consideration, for an Indian coal washery to produce clean coal containing not more than 17 % ash and a middling containing not more than 35 % ash. (These are the ash contents specified in India for coke ovens and thermal power plants respectively.) The simulation indicates that with such a flow sheet and the media densities/cut points specified (in lines 11230, 11290, 11300 and 11310) the yields of clean coal and middling would be 52.3 % and 29.5 % respectively.

To study the effect of the change of parameters like the cut points, the simulation can be repeated after changing the cut points (in columns 8 and 9 of line 11230 and column 8 of the other three lines mentioned above).

Figure 2 shows a flow sheet simulated for evaluating the differences in ash contents, if any, between the crusher products larger than the set and the other size fractions. The simulation results, including the echo printing of the data are given in appendix IV. The results indicate that there is not much difference in case of roll crusher (50 mm set), but there is appreciable difference (about 5 % ash) in case of the hammer mill.

Figure 3 shows a flow sheet that is designed to take some advantage of this phenomenon; the results (including the echo printing of the data) given in appendix V indicate that the +12.5 mm fraction of the product of the hammer mill (secondary crusher) may be sent direct to the middling product bunker and only the -12.5 mm fraction may be beneficiated. Simulation of the flow sheets in figures 2 and 3 also show the fraction of the oversize material in the crushers product.

### Conclusion and acknowledgement

This BASIC program is suitable for simulation and flow sheet calculations for the design of the coal beneficiation plants as well as for studying the effect of variation of the parameters like crusher sets, screen apertures, media densities etc. in an operational plant. It can be used in any personal computer or other small system having about 64 K memory. The memory requirement can be further reduced by changing the dimensions of the arrays for fewer size fractions, densities of cut, number of nodes and the number of flow streams. The capability of the program can be expanded, if the memory size of the computer permits, by adding more subroutines and by increasing the dimensions of the respective arrays.

The FORTRAN version of this program, for use in the main frame computers, was first developed between 1980 and 1982 at Indian Institute of Technology, Kharagpur under a research scheme sponsored by Central Mine Planning and Design Institute Ltd., Ranchi (India).

The author is grateful to Dr. Ing. P. Sen (now retired) Professor of I.I.T., Kharagpur who was the leader of the research project and to the authorities of C.M.P.D.I.L., Ranchi for financing the project.

The BASIC version described in this paper was developed at Indian School of Mines, Dhanbad. The author is grateful to Shri G.G. Sarkar, Scientist Emeritus and Dr. T.C. Rao the Head of the Department of Fuel and Mineral Engineering in Indian School of Mines for their help and guidance in the work.

### Selected references

- [1] Weyher, L.H.E. "Computer assistance in analysis and development of coal beneficiation flow sheets" - Mining Congress Journal, January 1971, page 39
- [2] Gotfried, B.S. "Computer analysis of coal preparation plant performance" - Proceedings of the 14th APCOM Symposium, Pennsylvania, 1976



- [3] Walter, A.D. and Ramani, B.V. "A review of computer oriented models of coal preparation plants" - 14th APCOM Symposium, Pennsylvania, 1976
- [4] Ball, P.J. and Tomlinson, R.C. "The use of computers to study the planning and control of coal preparation systems" - 15th APCOM Symposium, Brisbane, 1977
- [5] Crowe, C.L. et al "Chemical Plant Simulation" - Prentice Hall 1971
- [6] Himmelblau, D.M. and Bischoff, K.B. "Process analysis and simulation" - John Wiley and Sons 1968
- [7] Gotfried, B.S. and Jacobson, P.S. "A generalised distribution curve for characterising the performance of coal washing equipment"
- [8] Gotfried, B.S. et al "Computer simulation of coal preparation plants" - Final report no. FE-9000-1 November 1977.

Coal beneficiation flowsheet  
 Ref. line nos 11200-11350  
 of listing: Appendix I






Figure 1



Coal crushing flowsheet

Ref: Appendix IV

Figure 2



Coal crushing plant flow sheet

Ref: Appendix V

Figure 3

APPENDIX I

---

```

10 'PROGRAMME COALBEN FOR SIMULATION OF COAL BENEFICIATION FLOW SHEETS
20 'DEVELOPED BY PROFESSOR D.D..MISRA AT INDIAN SCHOOL OF MINES, DHANBAD
25 CLS: PRINT " SIMULATION IN PROGRESS"
30 DIM COAL(8,10,22),ASHF(8,10,22),MFS(20,7),EQUIP(20,5)
50 FOR K = 1 TO 22: FOR J = 1 TO 8: FOR I = 1 TO 5
60 COAL(I,J,K) = 0: ASHF(I,J,K) = 0: NEXT I: NEXT J: NEXT K
70 READ NSIZE,NDEN,K,FLWS,CMPs: LPRINT "SIZES" NSIZE " , DENSITIES " NDEN;
80 LPRINT " FEED FLOW STREAM NO." K: LPRINT " " FLWS,CMPs
90 FOR I = 1 TO NSIZE: READ SIZE(I): LPRINT SIZE(I);: NEXT I: LPRINT
100 FOR J = 1 TO NDEN: READ DEN(J): LPRINT DEN(J);: NEXT J: LPRINT
105 FOR I = 1 TO NSIZE
110 FOR J = 1 TO NDEN: READ COAL(I,J,K): LPRINT COAL(I,J,K);: NEXT J: LPRINT
120 FOR J = 1 TO NDEN: READ SH(I,J): LPRINT SH(I,J);
130 ASHF(I,J,K) = COAL(I,J,K)*SH(I,J): NEXT J: LPRINT: NEXT I
140 READ NP: LPRINT NP " FLOW SHEET MATRIX & EQUIPMENT DATA": FOR M = 1 TO NP
150 FOR N = 1 TO 7: READ MFS(M,N): LPRINT MFS(M,N);: NEXT N
160 FOR N = 1 TO 5: READ EQUIP(M,N): LPRINT EQUIP(M,N);: NEXT N
170 LPRINT: NEXT M: LPRINT: LPRINT: LPRINT "FEEDING OF DATA COMPLETED"
175 LPRINT "RESULTS OF SIMULATION FOLLOW": LPRINT: LPRINT: LPRINT
180 ITRN = 0: ODER = 1
190 ICON = 1: FOR NODE = ODER TO NP: KI = MFS(NODE,2)
200 K1 = MFS(NODE,5): K2 = MFS(NODE,6): K3 = MFS(NODE,7)
210 R1 = EQUIP(NODE,1): R2 = EQUIP(NODE,2): R3 = EQUIP(NODE,3)
220 IF MFS(NODE,1) = 0 THEN 250
225 IF MFS(NODE,3) > 0 THEN GOSUB 8020
230 ON MFS(NODE,1) GOSUB 1000,2000,3000,4000,5000,6000,7000,8000,9000,10000
240 IF ICON <> 1 THEN 190
245 GOTO 300
250 FOR N = 2 TO 7: KT = MFS(NODE,N): IF KT < 1 OR KT >30 THEN 300
260 Y = 0: AF = 0: FOR I = 1 TO NSIZE: FOR J = 1 TO NDEN
270 Y = Y+COAL(I,J,KT): AF = AF+ASHF(I,J,KT): NEXT J: NEXT I
280 AP = 0: IF Y > 0 THEN AP = AF/Y
290 LPRINT "STREAM NO."KT,FLWS;" " Y " "CMPs;AP
295 LPRINT: NEXT N
300 NEXT NODE
305 PRINT " SIMULATION COMPLETED"
310 STOP
990 '

```

```

1000 '*** SUBROUTINE CRUSHER FOR SIMULATION OF ROLLS AND IMPACTORS
1005 PRINT "          AT CRUSHER"
1030 EX = .91: IF R1 < 37.5 THEN EX = .85
1040 FOR J = 1 TO NDEN: JJ = J-1: DM = (DEN(J)+1.2)/2
1050 IF J > 1 THEN DM = (DEN(J)+DEN(JJ))/2
1060 A = 1.78/DM+0.05: IF R1 < 37.5 THEN A = .82/(DM-1.1)+.4
1070 FOR I = 1 TO NSIZE: B(I) = 1-EXP(-A*(SIZE(I)/R1)^EX)
1075 IF I = 1 THEN B(I) = 1
1080 NEXT I
1090 FOR I = 1 TO NSIZE: COAL(I,J,K1) = 0: ASHF(I,J,K1) = 0
1100 FOR L = 1 TO NSIZE: C(I,L) = 0: IF I = L THEN C(I,L) = 1
1110 II = I+1: IF SIZE(L) <= R1 THEN 1130
1120 C(I,L) = B(I): IF I < NSIZE THEN C(I,L) = B(I)-B(II)
1130 COAL(I,J,K1) = COAL(I,J,K1)+C(I,L)*COAL(L,J,K1)
1140 ASHF(I,J,K1) = ASHF(I,J,K1)+C(I,L)*ASHF(L,J,K1)
1150 NEXT L: NEXT I: NEXT J
1160 RETURN
1990 '
2000 '*** SUBROUTINE SCREEN FOR SIMULATION OF THE SCREENING OPERATION
2005 PRINT "          AT SCREEN "
2010 T = EQUIP(NODE,4): G = .387+.045*LOG(R1)
2020 FOR I = 1 TO NSIZE: II = I+1: D = SIZE(I)/2
2030 IF I < NSIZE THEN D = (SIZE(I)*SIZE(II))^0.5
2040 R = 1: IF D < R1 THEN R = EXP(-G*T*((R1-D)/(R1+R2))^R3)
2050 FOR J = 1 TO NDEN: COAL(I,J,K1) = COAL(I,J,K1)*R
2060 ASHF(I,J,K1) = ASHF(I,J,K1)*R: COAL(I,J,K2) = COAL(I,J,K1)-COAL(I,J,K1)
2070 ASHF(I,J,K2) = ASHF(I,J,K1)-ASHF(I,J,K1): NEXT J
2080 NEXT I: RETURN
2090 '
3000 '*** SUBROUTINE HMS FOR SIMULATION OF DENSE MEDIA BATHS & DRUMS
3005 PRINT "          AT H.M.S."
3010 RHO = R1: SK = .08: VI = .03: IF K3 = 0 THEN K3 = K2
3020 KF = K3: KS = K1: GOSUB 7020
3030 RHO = R2: KI = K1: KF = K2: KS = K1: IF K2 <> K3 THEN GOSUB 7020
3040 RETURN
3990 '
4000 '*** SUBROUTINE JIG WASHER FOR SIMULATION OF THE PNEUMATIC JIGS
4005 PRINT "          AT JIG"
4010 RHO = R1: SK = .0278: VI = .1: KS = K1: KF = K2: GOSUB 7020
4020 RHO = R2: KI = K2: KS = K2: KF = K3: IF KF > 0 THEN GOSUB 7020
4030 RETURN
4990 '
5000 '*** SUBROUTINE FOR SIMULATION OF H.M.CYCLONES & WATER ONLY CYCLONES
5005 PRINT "          AT CYCLONE"
5010 RHO = R1: SK = .012: VI = .04: IF RHO < 1.05 THEN VI = .35
5020 KS = K1: KF = K2: GOSUB 7020
5030 RETURN
5990 '
6990 '

```

```
7000 '*** SUBROUTINE FOR TROMP CURVE
7010 RHO = R1: SK = R2: VI = R3: KS = K1: KF = K2
7020 FOR I = 1 TO NSIZE: SM = SIZE(1)/2: II = 1+1
7030 IF SIZE(II) > 0 THEN SM = 2*SIZE(1)*SIZE(II)/(SIZE(I)+SIZE(II))
7040 IF SM > 0 THEN SS = RHO+10*SK/SM ELSE PRINT " SM = 0; STOP"
7045 EP = VI*(SS-1)
7050 FOR J = 1 TO NDEN: JJ = J-1: RJ = (DEN(J)+1.2)/2
7060 IF J > 1 THEN RJ = (DEN(J)+DEN(JJ))/2
7070 IF EP > 0 THEN X = (RJ-SS)/EP ELSE PRINT "EP = < 0; STOP"
7075 PV = 1: IF X > 2.59 THEN PV = 0
7080 IF X < 2.59 AND X > -2.59 THEN PV = .01*X^3-.26*X+.5
7090 COAL(I,J,KF) = COAL(I,J,KI)*PV: COAL(I,J,KS) = COAL(I,J,KI)-COAL(I,J,KF)
7100 ASHF(I,J,KF) = ASHF(I,J,KI)*PV: ASHF(I,J,KS) = ASHF(I,J,KI)-ASHF(I,J,KF)
7110 NEXT J: NEXT I: RETURN
7990 '
8000 '*** SUBROUTINE FOR BLENDING MORE THAN ONE FLOW STREAMS
8005 PRINT "          AT MIXING"
8010 RETURN
8020 IF MFS(NODE,1) < 1 THEN RETURN
8030 IF MFS(NODE,1) > 8 THEN RETURN
8040 NKI = 4: IF MFS(NODE,1) = 8 THEN NKI = 6
8045 KI = MFS(NODE,7): IF MFS(NODE,1) < 8 THEN KI = 21
8050 FOR I = 1 TO NSIZE: FOR J = 1 TO NDEN
8060 COAL(I,J,KI) = 0: ASHF(I,J,KI) = 0
8070 FOR N = 2 TO NKI: K = MFS(NODE,N): IF K < 1 THEN 8100
8080 COAL(I,J,KI) = COAL(I,J,KI)+COAL(I,J,K)
8090 ASHF(I,J,KI) = ASHF(I,J,KI)+ASHF(I,J,K)
8100 NEXT N
8110 NEXT J: NEXT I
8120 RETURN
8990 '
9000 '*** SUBROUTINE RECYCLE : CALCULATIONS BY DIRECT SUBSTITUTION
9005 PRINT "          AT RECYCLE"
9010 IF MFS(NODE,7) = 0 THEN MFS(NODE,7) = 50
9020 IF ITRN >= MFS(NODE,7) THEN 9120
9030 FOR J = 1 TO NDEN: FOR I = 1 TO NSIZE
9040 M1 = ABS(COAL(I,J,KI)-COAL(I,J,K1)): M2 = ABS(ASHF(I,J,KI)-ASHF(I,J,K1))
9050 IF M1 <= EQUIP(NODE,1) AND M2 <= EQUIP(NODE,2) THEN 9070
9060 COAL(I,J,K1) = COAL(I,J,KI): ASHF(I,J,K1) = ASHF(I,J,KI)
9065 ICON = 0
9070 NEXT I: NEXT J
9080 ITRN = ITRN+1
9090 IF ICON = 1 THEN ITRN = 0:
9100 IF ICON = 0 THEN ODER = MFS(NODE,4)
9110 RETURN
9120 LPRINT "NO CONVERGENCE IN" ITRN "ITERATIONS": ITRN = 0: ICON = 1
9125 LPRINT
9130 RETURN
9990 '
```

```
10000 'SUBROUTINE I : PRINTING DETAILED COMPOSITION OF A FLOW STREAM
10005 N = 2
10010 K = MFS(NODE,N): IF K < 1 THEN 10240
10020 LPRINT: LPRINT: LPRINT
10025 LPRINT "SIZE DENSITY DISTRIBUTION MATRIX OF STREAM" K
10030 FOR I = 1 TO NSIZE: FOR J = 1 TO NDEN
10040 IF COAL(I,J,K) THEN SH(I,J) = ASHF(I,J,K)/COAL(I,J,K)
10050 NEXT J: NEXT I
10060 FOR I = 1 TO NSIZE: DSF(I) = 0: CMW(I) = 0
10070 FOR J = 1 TO NDEN: DSF(I) = DSF(I)+ASHF(I,J,K)
10080 CMW(I) = CMW(I)+COAL(I,J,K): NEXT J
10090 IF CMW(I) > 0 THEN CSW(I) = DSF(I)/CMW(I)
10100 NEXT I
10110 FOR J = 1 TO NDEN: GSF(J) = 0: GMW(J) = 0
10120 FOR I = 1 TO NSIZE: GSF(J) = GSF(J)+ASHF(I,J,K)
10130 GMW(J) = GMW(J)+COAL(I,J,K): NEXT I
10140 IF GMW(J) > 0 THEN GSW(J) = GSF(J)/GMW(J)
10150 NEXT J: LPRINT
10160 LPRINT "DENSITY " ;: FOR I = 1 TO NSIZE
10170 LPRINT USING "###.###"; SIZE(I);: LPRINT "mm " ;: NEXT I
10180 LPRINT "TOTAL": FOR J = 1 TO NDEN
10190 LPRINT USING "###.###"; DEN(J);: LPRINT " " ;: FOR I = 1 TO NSIZE
10200 LPRINT USING "###.###"; COAL(I,J,K);: LPRINT " " ;: NEXT I
10210 LPRINT USING "###.###"; GMW(J): NEXT J
10220 LPRINT " TOTAL " ;: FOR I = 1 TO NSIZE
10230 LPRINT USING "###.###"; CMW(I);: LPRINT " " ;: NEXT I
10240 N = N+1: IF N < 8 THEN 10010
10250 RETURN
```



11000 DATA 8,10,1, FLOW RATE AS % OF FEED, WITH ASH %  
11010 DATA 200,100,50,25,12.5,6,1,.5  
11020 DATA 1.3,1.4,1.45,1.5,1.55,1.6,1.65,1.7,1.8,2.2  
11030 DATA 2.3,5,2.8,2.2,2.7,1.2,2.1,1.9,3.2,1.8  
11040 DATA 6.5,14.3,18.2,23.1,28.8,34.0,37.9,41.6,48.2,63.7  
11050 DATA 1.6,3.4,2.1,1.7,1.3,1.5,2.1,2.5,0.9,1.9  
11060 DATA 5.7,13.4,17.9,22.4,27.6,33.1,36.5,39.8,47.1,63.9  
11070 DATA 0.9,2.1,1.4,1.6,1.7,1.5,1.3,0.5,1.2,1.8  
11080 DATA 4.9,13.2,17.6,22.0,27.1,32.6,36.8,40.7,46.8,62.0  
11090 DATA 0.5,1.2,0.6,0.9,0.7,0.7,0.3,0.6,0.3,1.2  
11100 DATA 4.6,12.8,16.9,21.2,26.8,31.7,36.2,39.4,46.1,61.6  
11110 DATA 1.2,0.8,0.9,1.1,0.7,0.3,0.6,0.4,0.8,1.2  
11120 DATA 3.6,13.1,15.4,20.7,26.8,32.1,37.2,38.6,47.2,61.2,  
11130 DATA 1.3,1.6,0.7,1.4,0.6,0.4,0.5,1.1,0.9,2.5,  
11140 DATA 4.1,14.0,17.2,21.9,25.8,31.7,36.4,37.4,46.8,62.5  
11150 DATA 0.7,0.9,0.7,0.6,0.4,0.5,0.3,0.6,0.8,0.5  
11160 DATA 2.6,11.7,18.1,22.8,27.1,31.6,35.7,41.6,45.8,61.8  
11170 DATA 1.6,1.7,0.8,0.6,1.3,0.8,1.2,1.0,0.8,0  
11180 DATA 2.7,10.7,17.9,21.3,27.5,31.3,34.9,40.1,46.3,59.4  
11200 DATA 15  
11210 DATA 2,1,0,0,2,3,0,50,25,2,30,0  
11220 DATA 1,2,0,0,4,0,0,50,0,0,0,0  
11230 DATA 4,3,4,8,5,6,7,1.75,1.65,0,0,0  
11240 DATA 9,6,0,3,8,0,5,.1,.1,0,0,0  
11250 DATA 0,1,2,3,4,5,6,0,0,0,0,0  
11260 DATA 2,7,0,0,9,10,0,12.5,6,2,30,0  
11270 DATA 2,10,0,0,11,12,0,1,2,1,45,0  
11280 DATA 0,7,8,9,10,11,12,0,0,0,0,0  
11290 DATA 3,9,0,0,13,14,0,1.52,0,0,0,0  
11300 DATA 5,11,0,0,15,16,0,1.5,0,0,0,0  
11310 DATA 5,12,0,0,17,18,0,1,0,0,0,0  
11320 DATA 0,13,14,15,16,17,18,0,0,0,0,0  
11330 DATA 8,14,16,18,0,0,19,0,0,0,0,0  
11340 DATA 8,13,15,17,0,0,20,0,0,0,0,0  
11350 DATA 0,19,20,0,0,0,0,0,0,0,0,0

APPENDIX II

Data required for the different unit operations for coal washery simulation by program COALBEN

(The program ignores the parameters other than those given below. However to conform to the format of the flow sheet matrix for plant data some value, may be a zero, has to be given for each of the 12 elements of the row of the matrix)

Note - when a unit operation has less than the maximum permissible number of input streams the other input streams are numbered as zero. Sizes in mm, densities in gm/cc.

Unit operation	Column no.	Value to be assigned
Crusher	1	1
	2,3,4	Input stream numbers
	5	Output stream number
	6	Set of the crusher in mm
Screen	1	2
	2,3,4	Input stream number
	5	Oversize output stream number
	6	Undersize output stream number
	8	Screen aperture in mm
	9	Wire diameter in mm
10	1 for rectangular opening & 2 for square opening	
11	Residence time in seconds	
Dense media bath or drum	1	3
	2,3,4	Input stream numbers
	5,6,7	Output stream numbers in decreasing order of density
	8	Media density, first comptt.,
9	Media density, second comptt., if any	
Jig washers	1	4
	2,3,4	Input stream numbers in
	5,6,7	Output stream numbers in decreasing order of density
8,9	Densities of cut in the first and second compartments	

Dense media cyclones and water only cyclones	1 2,3,4 5,6 8	5 Input flow stream numbers Output stream numbers in decreasing order Media density gm/cc.
Blending	1 2,3,4,5,6 7	8 Input stream numbers Output stream number
Recycle	1 2 4 5 7 8,9	9 Input stream numbers Row number to which recycle takes place Output stream number Maximum number of iterations to be tried. Tolerances for CUAL & ASHF values
To print results	1 2,3,4,5,6,7	0 Stream numbers for which the results are to be printed.
To print the size density distribution matrix	1 2,3,4,5,6,7	10 Stream numbers for which compositions are to be printed.

APPENDIX III

SIZES 8 ,	DENSITIES 10						FEED FLOW STREAM NO. 1					
	FLOW RATE AS % OF FEED						WITH ASH %					
200	100	50	25	12.5	6	1	.5					
1.3	1.4	1.45	1.5	1.55	1.6	1.65	1.7	1.8	2.2			
2.3	5	2.8	2.2	2.7	1.2	2.1	1.9	3.2	1.8			
6.5	14.3	18.2	23.1	28.8	34	37.9	41.6	48.2	63.7			
1.6	3.4	2.1	1.7	1.3	1.5	2.1	2.5	.9	1.9			
5.7	13.4	17.9	22.4	27.6	33.1	36.5	39.8	47.1	63.9			
.9	2.1	1.4	1.6	1.7	1.5	1.3	.5	1.2	1.8			
4.9	13.2	17.6	22	27.1	32.6	36.8	40.7	46.8	62			
.5	1.2	.6	.9	.7	.7	.3	.6	.3	1.2			
4.6	12.8	16.9	21.2	26.8	31.7	36.2	39.4	46.1	64.6			
1.2	.8	.9	1.1	.7	.3	.6	.4	.8	1.2			
3.6	13.1	15.4	20.7	26.8	32.1	37.2	38.6	47.2	61.2			
1.3	1.6	.7	1.4	.6	.4	.5	1.1	.9	2.5			
4.1	14	17.2	21.9	25.8	31.7	36.4	37.4	46.8	62.5			
.7	.9	.7	.6	.4	.5	.3	.6	.8	.5			
2.6	11.7	18.1	22.8	27.1	31.6	35.7	41.6	45.8	61.8			
1.6	1.7	.8	.6	1.3	.8	1.2	1	.8	0			
2.7	10.7	17.9	21.3	27.5	31.3	34.9	40.1	46.3	59.4			

15 FLOW SHEET MATRIX & EQUIPMENT DATA

2	1	0	0	2	3	0	50	25	2	30	0
1	2	0	0	4	0	0	50	0	0	0	0
4	3	4	8	5	6	7	1.75	1.65	0	0	0
9	6	0	3	8	0	5	.1	.1	0	0	0
0	1	2	3	4	5	6	0	0	0	0	0
2	7	0	0	9	10	0	12.5	6	2	30	0
2	10	0	0	11	12	0	1	2	1	45	0
0	7	8	9	10	11	12	0	0	0	0	0
3	9	0	0	13	14	0	1.52	0	0	0	0
5	11	0	0	15	16	0	1.5	0	0	0	0
5	12	0	0	17	18	0	1	0	0	0	0
0	13	14	15	16	17	18	0	0	0	0	0
8	14	16	18	0	0	19	0	0	0	0	0
8	13	15	17	0	0	20	0	0	0	0	0
0	19	20	0	0	0	0	0	0	0	0	0

FEEDING OF DATA COMPLETED  
RESULTS OF SIMULATION FOLLOW

NO CONVERGENCE IN 5 ITERATIONS

STREAM NO. 1	FLOW RATE AS % OF FEED	100	WITH ASH % 29.0977
STREAM NO. 2	FLOW RATE AS % OF FEED	51.9258	WITH ASH % 29.3403
STREAM NO. 3	FLOW RATE AS % OF FEED	48.0742	WITH ASH % 28.8357
STREAM NO. 4	FLOW RATE AS % OF FEED	51.9258	WITH ASH % 29.3403
STREAM NO. 5	FLOW RATE AS % OF FEED	18.0895	WITH ASH % 55.1214
STREAM NO. 6	FLOW RATE AS % OF FEED	14.5547	WITH ASH % 43.1985
STREAM NO. 7	FLOW RATE AS % OF FEED	81.8025	WITH ASH % 23.3261
STREAM NO. 8	FLOW RATE AS % OF FEED	14.4466	WITH ASH % 43.2087
STREAM NO. 9	FLOW RATE AS % OF FEED	46.2803	WITH ASH % 22.755
STREAM NO. 10	FLOW RATE AS % OF FEED	35.5221	WITH ASH % 24.0701
STREAM NO. 11	FLOW RATE AS % OF FEED	19.7499	WITH ASH % 23.3739
STREAM NO. 12	FLOW RATE AS % OF FEED	15.7722	WITH ASH % 24.9419
STREAM NO. 13	FLOW RATE AS % OF FEED	12.7262	WITH ASH % 36.6482
STREAM NO. 14	FLOW RATE AS % OF FEED	33.5541	WITH ASH % 17.4857
STREAM NO. 15	FLOW RATE AS % OF FEED	6.16548	WITH ASH % 39.0588
STREAM NO. 16	FLOW RATE AS % OF FEED	13.5844	WITH ASH % 16.2551
STREAM NO. 17	FLOW RATE AS % OF FEED	10.5613	WITH ASH % 29.0612
STREAM NO. 18	FLOW RATE AS % OF FEED	5.21097	WITH ASH % 16.5931
STREAM NO. 19	FLOW RATE AS % OF FEED	52.3495	WITH ASH % 17.0775
STREAM NO. 20	FLOW RATE AS % OF FEED	29.4529	WITH ASH % 34.4323

APPENDIX IV

SIZES 8 ,	DENSITIES 10	FEED FLOW STREAM NO. 1									
FLOW RATE AS % OF FEED											WITH ASH %
200	100	50	25	12.5	6	3	1.5	0.75	0.375	0.1875	0.09375
1.3	1.4	1.45	1.5	1.55	1.6	1.65	1.7	1.8	2.2		
2.3	5	2.8	2.2	2.7	1.2	2.1	1.9	3.2	1.8		
6.5	14.3	18.2	23.1	28.8	34	37.9	41.6	48.2	63.7		
1.6	3.4	2.1	1.7	1.3	1.5	2.1	2.5	.9	1.9		
5.7	13.4	17.9	22.4	27.6	33.1	36.5	39.8	47.1	63.9		
.9	2.1	1.4	1.6	1.7	1.5	1.3	.5	1.2	1.8		
4.9	13.2	17.6	22	27.1	32.6	36.8	40.7	46.8	62		
.5	1.2	.6	.9	.7	.7	.3	.6	.3	1.2		
4.6	12.8	16.9	21.2	26.8	31.7	36.2	39.4	46.1	64.6		
1.2	.8	.9	1.1	.7	.3	.6	.4	.8	1.2		
3.6	13.1	15.4	20.7	26.8	32.1	37.2	38.6	47.2	61.2		
1.3	1.6	.7	1.4	.6	.4	.5	1.1	.9	2.5		
4.1	14	17.2	21.9	25.8	31.7	36.4	37.4	46.8	62.5		
.7	.9	.7	.6	.4	.5	.3	.6	.8	.5		
2.6	11.7	18.1	22.8	27.1	31.6	35.7	41.6	45.8	61.8		
1.6	1.7	.8	.6	1.3	.8	1.2	1	.8	0		
2.7	10.7	17.9	21.3	27.5	31.3	34.9	40.1	46.3	59.4		

9 FLOW SHEET MATRIX & EQUIPMENT DATA

2	1	0	0	2	3	0	50	25	2	30	0
1	2	0	0	4	0	0	50	0	0	0	0
2	4	0	0	5	6	0	50	25	2	30	0
2	6	0	0	7	8	0	12.5	6	2	30	0
1	7	0	0	9	0	0	12.5	0	0	0	0
2	9	0	0	10	11	0	12.5	6	2	30	0
2	11	0	0	12	13	0	1	2	1	25	0
0	2	3	4	5	6	7	0	0	0	0	0
0	8	9	10	11	12	13	0	0	0	0	0

FEEDING OF DATA COMPLETED  
RESULTS OF SIMULATION FOLLOW

STREAM NO. 2	FLOW RATE AS % OF FEED	51.9258	WITH ASH %	29.3403
STREAM NO. 3	FLOW RATE AS % OF FEED	48.0742	WITH ASH %	28.8357
STREAM NO. 4	FLOW RATE AS % OF FEED	51.9258	WITH ASH %	29.3403
STREAM NO. 5	FLOW RATE AS % OF FEED	23.2763	WITH ASH %	30.6803
STREAM NO. 6	FLOW RATE AS % OF FEED	28.6495	WITH ASH %	28.2516
STREAM NO. 7	FLOW RATE AS % OF FEED	19.1241	WITH ASH %	28.6485
STREAM NO. 8	FLOW RATE AS % OF FEED	9.52535	WITH ASH %	27.4548
STREAM NO. 9	FLOW RATE AS % OF FEED	19.1241	WITH ASH %	28.6485
STREAM NO. 10	FLOW RATE AS % OF FEED	4.69653	WITH ASH %	34.0346
STREAM NO. 11	FLOW RATE AS % OF FEED	14.4276	WITH ASH %	26.8952
STREAM NO. 12	FLOW RATE AS % OF FEED	10.2184	WITH ASH %	28.2189
STREAM NO. 13	FLOW RATE AS % OF FEED	4.20921	WITH ASH %	23.6818

SIZES 8 , DENSITIES 10 FEED FLOW STREAM NO. 1

FLOW RATE AS % OF FEED WITH ASH %

200	100	50	25	12.5	6	1	.5						
1.3	1.4	1.45	1.5	1.55	1.6	1.65	1.7	1.8	2.2				
2.3	5	2.8	2.2	2.7	1.2	2.1	1.9	3.2	1.8				
6.5	14.3	18.2	23.1	28.8	34	39.9	41.6	48.2	63.7				
1.6	3.4	2.1	1.7	1.3	1.5	2.1	2.5	.9	1.9				
5.7	13.4	17.9	22.4	27.6	33.1	36.5	39.8	47.1	63.9				
.9	2.1	1.4	1.6	1.7	1.5	1.3	.5	1.2	1.8				
4.9	13.2	17.6	22	27.1	32.6	36.8	40.7	46.8	62				
.5	1.2	.6	.9	.7	.7	.3	.6	.3	1.2				
4.6	12.8	16.9	21.2	26.8	31.7	36.2	39.4	46.1	64.6				
1.2	.8	.9	1.1	.7	.3	.6	.4	.8	1.2				
3.6	12.1	15.4	20.7	26.8	32.1	37.2	38.6	47.2	61.2				
1.3	1.6	.7	1.4	.6	.4	.5	1.1	.9	2.5				
4.1	14	17.2	21.9	25.8	31.7	36.4	37.4	46.8	62.5				
.7	.9	.7	.6	.4	.5	.3	.6	.8	.5				
2.6	11.7	18.1	22.8	27.1	31.6	35.7	41.6	45.8	61.8				
1.6	1.7	.8	.6	1.3	.8	1.2	1.3	1.8	0				
2.7	10.7	17.9	21.3	27.5	31.9	34.9	40.1	46.3	59.4				

8 FLOW SHEET MATRIX & EQUIPMENT DATA

2	1	0	0	2	3	0	50	25	2	30	0		
1	2	0	0	4	0	0	50	0	0	0	0		
2	3	4	0	5	6	0	12.5	6	2	30	0		
1	5	0	0	7	0	0	12.5	0	0	0	0		
2	7	0	0	8	9	0	12.5	6	2	30	0		
2	6	9	0	10	11	0	1	2	1	75	0		
0	1	2	3	4	5	6	0	0	0	0	0		
0	7	8	9	10	11	0	0	0	0	0	0		

FEEDING OF DATA COMPLETED  
RESULTS OF SIMULATION FOLLOW

STREAM NO. 1	FLOW RATE AS % OF FEED	100	WITH ASH %	29.0977
STREAM NO. 2	FLOW RATE AS % OF FEED	51.9258	WITH ASH %	29.3403
STREAM NO. 3	FLOW RATE AS % OF FEED	48.0742	WITH ASH %	28.8357
STREAM NO. 4	FLOW RATE AS % OF FEED	51.9258	WITH ASH %	29.3403
STREAM NO. 5	FLOW RATE AS % OF FEED	60.0429	WITH ASH %	29.8217
STREAM NO. 6	FLOW RATE AS % OF FEED	39.9572	WITH ASH %	28.0098
STREAM NO. 7	FLOW RATE AS % OF FEED	60.0429	WITH ASH %	29.8217
STREAM NO. 8	FLOW RATE AS % OF FEED	14.5175	WITH ASH %	35.771
STREAM NO. 9	FLOW RATE AS % OF FEED	45.5254	WITH ASH %	27.9245
STREAM NO. 10	FLOW RATE AS % OF FEED	55.2629	WITH ASH %	29.5713
STREAM NO. 11	FLOW RATE AS % OF FEED	30.2196	WITH ASH %	25.0258